Lecture 14

IIR filters 3

Introduction

In the previous lecture we learned how to convert a lowpass prototype filter into an IIR lowpass or highpass filter. In this lecture we extend that technique to bandpass and bandstop filters.

Bandpass filter

Previously we derived the lowpass-to-bandpass filter transformation

$$\tilde{s}=\frac{s^2+\Omega_l\Omega_h}{(\Omega_h-\Omega_l)s} \tag{1}$$

We also derived the discrete-to-continuous frequency transformation

$$s=\frac{2}{T}\frac{1-z^{-1}}{1+z^{-1}} \tag{2}$$

Now we want to combine (1) and (2) into a transformation that converts a lowpass prototype filter into an IIR bandpass filter with band frequencies f_l and f_h. It will be convenient to break (2) up as

$$\begin{aligned} s&=\frac{2}{T}u \\ u&=\frac{1-z^{-1}}{1+z^{-1}} \end{aligned} \tag{3}$$

For $z=e^{j\omega}=e^{j2\pi f}$ we have

$$u=\frac{1-e^{-j2\pi f}}{1+e^{-j2\pi f}}=\frac{e^{j\pi f}-e^{-j\pi f}}{e^{j\pi f}+e^{-j\pi f}}=j\tan(\pi f) \tag{4}$$

Let's write (1) as

$$\tilde{s}=\frac{\left(\frac{2}{T}\right)^2u^2+\Omega_l\Omega_h}{(\Omega_h-\Omega_l)\left(\frac{2}{T}\right)u} \tag{5}$$

Multiplying numerator and denominator by $(T/2)^2$ we get

$$\tilde{s}=\frac{u^2+\left(\frac{T}{2}\right)^2\Omega_l\Omega_h}{\left(\frac{T}{2}\right)(\Omega_h-\Omega_l)u} \tag{6}$$

Let's write this as

$$\tilde{s}=\frac{u^2+A}{Bu} \tag{7}$$

The lowpass prototype passband is $-1\leq\tilde{\Omega}\leq 1$ (Fig. 1). We want the discrete passband $f_l \leq f \leq f_h$ to map to this.



Fig. 1: Lowpass prototype frequency response.

Substituting $\tilde{s}=-j$ and $u=j\tan(\pi f_l)$ into (7) we obtain

$$-j=\frac{-\tan^2(\pi f_l)+A}{B\, j\tan(\pi f_l)} \tag{8}$$

Clearing fractions results in

$$B\tan(\pi f_l)=-\tan^2(\pi f_l)+A \tag{9}$$

Then substituting $\tilde{s}=j$ and $u=j\tan(\pi f_h)$ into (7) we obtain

$$j=\frac{-\tan^2(\pi f_l)+A}{B\, j\tan(\pi f_l)} \tag{10}$$

Clearing fractions results in

$$-B\tan(\pi f_h)=-\tan^2(\pi f_h)+A \tag{11}$$

The solution of (9) and (11) is

$$\begin{aligned} A&=\tan(\pi f_l)\tan(\pi f_h)\\ B&=\tan(\pi f_h)-\tan(\pi f_l)\end{aligned} \tag{12}$$

With this (7) becomes

$$\tilde{s}=\frac{u^2+\tan(\pi f_l)\tan(\pi f_h)}{[\tan(\pi f_h)-\tan(\pi f_l)]u} \tag{13}$$

and the desired IIR bandpass transfer function is

$$\boxed{H(z)=H_p\left(\frac{u^2+\tan(\pi f_l)\tan(\pi f_h)}{[\tan(\pi f_h)-\tan(\pi f_l)]u}\right)} \tag{14}$$

with

$$u=\frac{1-z^{-1}}{1+z^{-1}} \tag{15}$$



Fig. 2: 4th order IIR bandpass filter with $f_l = 0.2, f_h = 0.3$.

An example is shown in Fig. 2.

Bandstop filter

lowpass-to-bandstop filter transformation is simply the inverse of (1). It follows that the desired IIR bandstop transfer function is

$$\boxed{H(z) = H_p\left(\frac{[\tan(\pi f_h) - \tan(\pi f_l)]\, u}{u^2 + \tan(\pi f_l)\tan(\pi f_h)}\right)} \tag{16}$$

with

$$u = \frac{1 - z^{-1}}{1 + z^{-1}} \tag{17}$$

An example is shown in Fig. 3.



Fig. 3 8th order IIR bandstop filter with $f_l=0.2, f_h=0.3$.

Appendix

```
function [b,a] = BPiir(fl,fh,n) //bandpass IIR filter coefficients
  w = poly(0,'w');              //based on n-th order Butterworth
  u = (1-w)/(1+w);
  th = tan(%pi*fh);
  tl = tan(%pi*fl);
  s = (u^2+tl*th)/((th-tl)*u);
  B = Butterworth(n);
  H = 1/horner(B,s);
  a = coeff(H(3));
  b = coeff(H(2))/a(1);
  a = a/a(1);
endfunction

function [b,a] = BSiir(fl,fh,n) //bandstop IIR filter coefficients
  w = poly(0,'w');              //based on n-th order Butterworth
  u = (1-w)/(1+w);
  th = tan(%pi*fh);
  tl = tan(%pi*fl);
  s = ((th-tl)*u)/(u^2+tl*th);
  B = Butterworth(n);
  H = 1/horner(B,s);
  a = coeff(H(3));
  b = coeff(H(2))/a(1);
  a = a/a(1);
endfunction
```